
This lecture

1 2 3 4 5 6 7 8 9 10 11 12 13 14 15

I · what a proof is II · how proofs are built III · correct programs IV · AI in the loop

↓ you are here theory in Lean

Absurdity

⊥ := α: ∗ . α,
¬A := A→⊥

False, Not

▶ A false
proposition is an
empty type, and
from an
inhabitant of the
empty type
anything follows.

1

Connectives

∧, ∨,⇔ as
-types over all
propositions

And, Or, Iff

▶ The rules of a
connective are
theorems
derived from its
coding, not
axioms.

2

Existence

∃ := α: ∗
. ((:S. P→α)→
α)

Exists, 〈w, h〉

▶ A proof of ∃
carries a witness,
and using it
means raising a
flag for that
witness.

3

Classical

the excluded third
as an assumed
inhabitant ET

Classical.em

▶ Excluded
middle is not
derivable; it is
assumed, and
every proof that
uses it says so.

4

Goals

the goal first:
? : A, then the
lines above it

by, intro, apply,
exact

▶ A proof is
found from its
goal upward, and
the machine
keeps the flags
and the holes.

5

History: the rules of logic, and where the extra one enters

L. E. J. Brouwer
1881–1966

A proof of A∨ B
must say which;

the excluded third
is not a law.

1908 Brouwer rejects the excluded third as a general
law of reasoning.

1930 Heyting formalises the logic that results: intu-
itionistic, today called constructive.

1935 Gentzen: natural deduction, one introduction
and one elimination rule per connective.

1969 Howard: the correspondence, extended to ∧, ∨,
⊃, ∀, ∃.

1972 Girard: system F, where the connectives, long
known to be definable from ∀ and ⇒, become
λ-terms.

1985 Coquand and Huet: the same codings in λC. This
lecture’s chapter.

Gerhard
Gentzen
1909–1945

Each connective
has a rule to prove
it and a rule to use

it.
Constructive logic is what the codings give for free; classical logic is one extra assumption, and this
lecture shows where it enters. Photos: E. Menzler-Trott/MFO (CC BY-SA 2.0 de) and Wikimedia
Commons. N&G §7.4, §7.8

Minimal logic, and what is missing

Natural deduction for predicate logic has seven operations; λC has one type
former, . The first two are -types with rules (pp) and (bst); the other five
are -types too, and their rules are derived.

logic coded in λC its rules

A⇒ B A→B (pp), (bst)
∀∈S (P()) :S. P (pp), (bst)

⊥ ? derived
¬A ? derived
A∧ B ? derived
A∨ B ? derived
∃∈S (P()) ? derived

Derived means: the introduction and elimination rules of natural deduction
become derivable rules of λC, and each derivation is a term. N&G §7.1

Absurdity: from the property to the type

What characterises ⊥ in logic: if ⊥ holds, every proposition holds.

Definition. ⊥ := α: ∗ . α

A product over all propositions: a
λ2 type.

⊢ ∗ : □ α : ∗ ⊢ α : ∗

⊢ α: ∗ . α : ∗
(ƒorm), pair (□,∗)

The domain ∗ has sort □; the body α has sort ∗. λP
has no (□,∗), so ⊥ cannot be formed there.

N&G §7.1, Rem. 7.1.1

⊥-elimination is one application

An inhabitant ƒ of ⊥ is a function on propositions: give it any A : ∗ and it
returns a proof of A.

⊥

A
(⊥-elim)

() ƒ : α: ∗ . α

() ƒ A : A (pp) on (), A : ∗

The rule is not postulated. It is (pp) with a proposition as the argument.
N&G §7.1, Rem. 7.1.1

Negation

Definition
¬A := A→⊥, that is, A⇒⊥.

The rules of ¬ are the rules of ⇒ with ⊥ for B:

Assume A · · · ⊥

¬A
(¬-intro)

Assume A · · · B

A⇒ B
(⇒-intro)

¬A A

⊥
(¬-elim)

A⇒ B A

B
(⇒-elim)

Instance, with A : ∗,  : A→⊥,  : A:  : ⊥ by (pp). This is (¬-elim), and
also (⊥-intro).
Not an instance: ¬S for a set S : ∗, although S→⊥ is a well-formed type.
Next slide. N&G §7.1, Rem. 7.1.2

False and Not, in Lean

-- §7.1: ⊥ := α:∗. α, and ⊥-elim is one (appl)
def Bot : Prop := ∀ α : Prop, α
theorem bot_elim (f : Bot) (A : Prop) : A := f A

-- Lean: False has no constructor; Not is the same abbreviation
#print False -- inductive False : Prop (no constructors)
#print Not -- def Not : Prop → Prop := fun a => a → False
#check @False.elim -- {C : Sort u_1} → False → C
#check @absurd -- {a : Prop} → {b : Sort u_1} → a → ¬a → b

-- the two readings of "empty" agree, in both directions
example : False → Bot := fun h _ => h.elim
example : Bot → False := fun f => f False

Two readings of “empty”: False has no constructor, the coded ⊥ has no closed
inhabitant. The two terms at the bottom show they are the same proposition.

A set is not a statement

In λC, N and 2 = 3 both have sort ∗, so ¬ applies to both. Lean splits ∗ into
Type and Prop, and ¬ applies to Prop only.

#check (2 = 3) → False -- 2 = 3 → False : Prop ¬(2 = 3)
#check Nat → False -- ∀ (a : Nat), False : Prop "Nat is empty": a proposition too
example : (Nat → False) → False := fun h => h 0 -- and 0 refutes it
#check_failure ¬ Nat -- Nat has type Type but is expected to have type Prop

N→⊥ is still a proposition, “N is empty”; what Lean refuses is to call it ¬N.

Conjunction

Definition. A∧ B := C: ∗ . (A→B→C)→C. Read  as “for all” and→as
“implies”:

for all C: if (A and B together) imply C, then C.

The condition “A and B together” is redundant exactly when both hold. So the
sentence says that both hold, and says it without a ∧.

A B

A∧ B
(∧-intro)

A B

C: ∗ . (A→B→C)→C
(∧-intro-sec)

A∧ B

A
(∧-elim-left)

C: ∗ . (A→B→C)→C

A
(∧-elim-left-sec)

The right-hand rules are the claim. They hold if, under the coding, a term of the
conclusion’s type can be built from terms of the premisses’ types. N&G §7.2, I

∧-intro-sec, derived

Find ?1 : C: ∗ . (A→B→C)→C from  : A and y : B. Bottom to top: the  opens
a flag for C, the→a flag for z.

() A : ∗ B : ∗
(c)  : A y : B

(e) C : ∗
(ƒ ) z : A→B→C
(1) z  : B→C (pp) on (ƒ ), (c)
(2) z y : C (pp) on (1), (c)
(3) λz:A→B→C. z y : (A→B→C)→C (bst) on (2)
(4) λC: ∗ . λz:A→B→C. z y : C: ∗ . (A→B→C)→C (bst) on (3)

And the eliminations, in one line each, with  : A∧ B: A (λ:A.λ:B.) : A
and B (λ:A.λ:B.) : B.

N&G §7.2, Ex. 7.4

And, in Lean

The coding, and its three rules derived.
-- A ∧ B := C:∗. (A → B → C) → C
def And2 (A B : Prop) : Prop := ∀ C : Prop, (A → B → C) → C

theorem And2.intro {A B : Prop} (x : A) (y : B) : And2 A B := fun _ z => z x y
theorem And2.left {A B : Prop} (u : And2 A B) : A := u A (fun x _ => x)
theorem And2.right {A B : Prop} (u : And2 A B) : B := u B (fun _ y => y)

Lean’s And: an inductive type, the same three rules. The constructor is
∧-intro, the two fields are the eliminations.
#print And -- structure And (a b : Prop) : Prop

-- fields: And.left : a And.right : b
#check @And.intro -- ∀ {a b : Prop}, a → b → a ∧ b
example (A B : Prop) (h : A ∧ B) : B ∧ A := 〈h.right, h.left〉

Every connective of this lecture is inductive in Lean; what that buys over the coding is a
later lecture’s topic.

Disjunction

Definition. A∨ B := C: ∗ . (A→C)→ (B→C)→C.

for all C: if A implies C and B implies C, then C.

Two cases, and in each C follows: that is what a proof from A∨ B looks like. The
coding is the elimination rule, spelled out.

A

A∨ B
(∨-intro-left)

B

A∨ B
(∨-intro-right)

A∨ B A⇒ C B⇒ C

C
(∨-elim)

Instances, with  : A and  : A∨ B,  : A→C,  : B→C:
λC: ∗ . λ:A→C.λ:B→C.  : A∨ B and C : C.
Not an instance: C does not prove C. Its type is (B→C)→C: the case B is
still owed, and the type says so. N&G §7.2, II

The coding says exactly ∨: both directions

Why this -type and no other: A∨ B and its coding imply each other, and the
argument each way is one step.

N&G §7.2, II

∨-elim-sec, derived

Find ? : C from  : A∨ B, y : A→C, z : B→C. Under the coding,  expects a
proposition and two functions: three applications.

() A : ∗ B : ∗ C : ∗

(d)  : D: ∗ . (A→D)→ (B→D)→D

(e) y : A→C z : B→C

(1) C : (A→C)→(B→C)→C (pp) on (d), ()

(2) Cy : (B→C)→C (pp) on (1), (e)

(3) Cy z : C (pp) on (2), (e)

In (1) the argument C replaces the bound D in the type of : the (pp) of λ2.
The introductions are Exercise 7.7: open the coding’s three flags, write   : C,
close them with three (bst). N&G §7.2, p. 143

Or, in Lean

-- A ∨ B := C:∗. (A → C) → (B → C) → C
def Or2 (A B : Prop) : Prop := ∀ C : Prop, (A → C) → (B → C) → C
theorem Or2.inl {A B : Prop} (x : A) : Or2 A B := fun _ f _ => f x
theorem Or2.elim {A B C : Prop} (x : Or2 A B) (y : A → C) (z : B → C) : C :=

x C y z

#print Or -- inductive Or : Prop → Prop → Prop
-- Or.inl : a → a ∨ b Or.inr : b → a ∨ b

#check @Or.elim -- ∀ {a b c : Prop}, a ∨ b → (a → c) → (b → c) → c

-- ∨-elim, then ∨-intro-right and ∨-intro-left
example (A B : Prop) (h : A ∨ B) : B ∨ A :=

h.elim (fun a => Or.inr a) (fun b => Or.inl b)
example (A B : Prop) (h : A ∨ B) : B ∨ A := -- the same, by pattern

match h with
| .inl a => .inr a
| .inr b => .inl b

Or.elim is ∨-elim-sec with its three propositions made implicit. match is the same rule
read as case analysis: one arm per constructor.

Biimplication

Definition
A⇔B := (A⇒ B)∧ (B⇒ A).

Nothing to derive: the rules of⇔ are the rules of ∧ applied to two implications.
N&G Rem. 7.3.1

-- Rem. 7.3.1: A ⇔ B := (A ⇒ B) ∧ (B ⇒ A). Lean: a structure with
two fields.

#print Iff -- structure Iff (a b : Prop) : Prop
-- fields: Iff.mp : a → b Iff.mpr : b → a

#check @Iff.intro -- ∀ {a b : Prop}, (a → b) → (b → a) → (a ↔ b)

example (A B : Prop) (h : A ↔ B) (a : A) : B := h.mp a
example (A B : Prop) : (A ∧ B) ↔ (B ∧ A) :=
〈fun h => 〈h.2, h.1〉, fun h => 〈h.2, h.1〉〉

A proof in propositional logic: (A∨ B)⇒ (¬A⇒ B)

() A : ∗ B : ∗
(c)  : C: ∗ . ((A→C)→(B→C)→C)
(d) y : A→⊥
(1) B : (A→B)→(B→B)→B (pp) on (c), ()
(e)  : A
(2) y : ⊥ (pp) on (d), (e)
(3) yB : B (pp) on (2), ()
(4) λ:A. y B : A→B (bst) on (3)
(5) B (λ:A. y B) : (B→B)→B (pp) on (1), (4)
(ƒ )  : B
(6)  : B (r) on (ƒ )
(7) λ:B. : B→B (bst) on (6)
(8) B (λ:A. y B) (λ:B.) : B (pp) on (5), (7)
(9) λy:A→⊥. B (λ:A. y B) (λ:B.) : (A→⊥)→B (bst) on (8)
(10) λ:A∨ B.λy:A→⊥. . . . : (A∨ B)→(A→⊥)→B (bst) on (9)

The one step that needs an idea is (1): apply  to the goal B itself, since the
result then ends in B. N&G §7.3, Fig. 7.1

Computational Logic in the Era of AI 18 / 43



The same proof, as one term

-- the proof object of (A ∨ B) ⇒ (¬A ⇒ B), line (10) of the derivation
theorem or_neg (A B : Prop) : A ∨ B → ¬A → B :=

fun x y => x.elim (fun u => (y u).elim) (fun v => v)

#print or_neg
-- theorem or_neg : ∀ (A B : Prop), A ∨ B → ¬A → B :=
-- fun A B x y => Or.elim x (fun u => False.elim (y u)) fun v => v

line of the derivation subterm rule

(2), (3) (y u).elim (pp) twice; ⊥-elim is .elim
(4) fun u => (y u).elim (bst)
(7) fun v => v (bst)
(1), (5), (8) x.elim (...) (...) (pp) three times: ∨-elim
(9), (10) fun x y => ... (bst) twice

The term is the definiens of line (10), and Lean checks it against the proposition.

Classical logic

Everything so far is constructive logic. Classical logic has two more laws, and
neither is derivable from the rules given so far:

ET: A∨¬A DN: ¬¬A⇒ A.

Constructive logic with ET derives DN, and with DN derives ET, so adding one
suffices.

How to add ET to λC
Not a new rule: an assumed inhabitant, as a flag in front of every context.

ET : α: ∗ . α∨¬α

An axiom is a variable of the proposition’s type, declared once and never
abstracted away.

Every proof under that flag is classical, and says so: ET occurs in its term. A
proof that does not mention ET is constructive. N&G §7.4

ET gives DN: the derivation, from the goal

Goal: ? : β: ∗ .¬¬β→β under the flag ET . Two flags open, and ET ββ leaves
two implications to prove.

() ET : α: ∗ . α∨¬α
(b) β : ∗
(c)  : ¬¬β
(1) ET β : β∨¬β (pp) on (), (b)
(2) ET β : γ: ∗ . (β→γ)→ (¬β→γ)→γ the coding of ∨, unfolded
(3) ET ββ : (β→β)→(¬β→β)→β (pp) on (2), (b)
(4) λy:β. y : β→β (bst)
(5) ET ββ (λy:β. y) : (¬β→β)→β (pp) on (3), (4)
(d) z : ¬β
(6) z : ⊥ (pp) on (c), (d)
(7) z β : β (pp) on (6), (b)
(8) λz:¬β.  z β : ¬β→β (bst) on (7)
(9) ET ββ (λy:β. y) (λz:¬β.  z β) : β (pp) on (5), (8)

Two more (bst) close (c) and (b). Line (2) is (con): the abbreviation ∨ is
unfolded so that (pp) can see the . N&G §7.4

The excluded third, in Lean

As a hypothesis: the flag ET is a parameter. The proof depends on no
axiom; it is constructive relative to em.
-- the excluded third as a flag in front of the context: a parameter em
theorem dn (q : Prop) (em : ∀ p : Prop, p ∨ ¬p) (h : ¬¬q) : q :=

(em q).elim (fun y => y) (fun z => (h z).elim)
#print axioms dn -- ’dn’ does not depend on any axioms

As a library constant: Classical.em discharges the flag. Now the proof
depends on three axioms, and none is called “em”.
#check Classical.em -- Classical.em (p : Prop) : p ∨ ¬p
#check @Classical.byContradiction -- ∀ {p : Prop}, (¬p → False) → p
theorem dn’ (q : Prop) (h : ¬¬q) : q := dn q Classical.em h
#print axioms dn’
-- ’dn’’ depends on axioms: [propext, Classical.choice, Quot.sound]

Lean does not assume the excluded third; it proves it from choice, propositional
extensionality and quotients. One flag on the board, three in the kernel: what they are
is next lecture.

Existence

∀ is  itself. For ∃ the first-order definition ¬∀∈S (¬P()) holds only classically;
the second-order coding works in constructive logic.
Definition. ∃∈S (P()) := α: ∗ . ((:S. P→α)→α).

for all α: if for every  in S, P() implies α, then α.

That is the elimination rule, and it says how an existence is used:

 ∈ S P()

∃∈S (P())
(∃-intro)

∃∈S (P()) ∀∈S (P()⇒ A)

A
(∃-elim)  /∈ FV(A)

The mathematician’s shortcut: “there is an  with P(); let  be such that
P() holds; . . . ; hence A.” The  of the second sentence has a scope, and A
must not mention it. The rule makes that scope a flag. N&G §7.5, Rem. 7.5.1

∃-elim-sec and ∃-intro-sec, derived

Elimination. From y, an existence, and
z, a function on witnesses: two
applications.

() S : ∗ P : S→∗ A : ∗

(d) y : α: ∗ . ((:S. P→α)→α)

(e) z : :S. P→A

(1) yA : (:S. P→A)→A on (d), ()

(2) yAz : A on (1), (e)

Introduction. From  : S and
 : P, build a function of α and :

(c)  : S  : P

(e) α : ∗

(ƒ )  : :S. P→α

(1)   : α on (ƒ ), (c)

(2) λ: . . . .   : (. . . )→α (bst)

(3) λα: ∗ . λ: . . . .   : ∃∈S(P)(bst)

Why  /∈ FV(A) is essential: in line (1) of the elimination, A is substituted for α
in a type where  is bound. If A mentioned , that  would be captured, the
defect L1’s substitution was built to avoid. N&G §7.5, Rem. 7.5.2, Ex. 7.11

Exists, in Lean

-- ∃x:S. P x := α:∗. ((x:S. P x → α) → α)
def Ex2 (S : Type) (P : S → Prop) : Prop :=
∀ α : Prop, (∀ x : S, P x → α) → α

theorem Ex2.intro {S : Type} {P : S → Prop} (a : S) (u : P a) : Ex2 S P :=
fun _ v => v a u

theorem Ex2.elim {S : Type} {P : S → Prop} {A : Prop}
(y : Ex2 S P) (z : ∀ x, P x → A) : A := y A z

#print Exists -- inductive Exists : (α → Prop) → Prop
-- Exists.intro (w : α) : p w → Exists p

example : ∃ n : Nat, n + n = 6 := 〈3, rfl〉 -- the witness, then the proof
example (S : Type) (P : S → Prop) (h : ∃ x, P x) (k : ∀ x, P x → False) :

False :=
match h with -- ∃-el: open the flag
| 〈w, hw〉 => k w hw

〈3, rfl〉 is ∃-intro-sec with the witness first; match is ∃-elim-sec, and its arm is the flag
 : S,  : P of ∃-elim.

Computational Logic in the Era of AI 25 / 43



The witness stays inside the flag

-- the conclusion A may not mention the witness x.
-- Lean’s version of the side condition is a sort: an Exists in Prop may
-- only be eliminated into a Prop, so no function can return the witness.
#guard_msgs(drop error) in
def witness (h : ∃ n : Nat, n + n = 6) : Nat :=

match h with
| 〈w, _〉 => w

-- error: recursor ’Exists.casesOn’ can only eliminate into Prop

theorem witness’ (h : ∃ n : Nat, n + n = 6) : ∃ m : Nat, m + m + 2 = 8 :=
match h with -- into Prop: accepted
| 〈w, hw〉 => 〈w, by omega〉

The side condition  /∈ FV(A) keeps the witness out of the conclusion. Lean enforces it
by sort: an Exists in Prop is eliminated into a Prop only. The full reason comes with
inductive types.

A proof in predicate logic: ¬∃∈S (P())⇒ ∀y∈S (¬P(y))

Three flags for the three leading binders; then a ⊥ is wanted, and the only way
to one is , applied to an existence built from y and .

() S : ∗ P : S→∗

(c)  : ¬(∃:S. P)

(d) y : S

(e)  : Py

(1) λα: ∗ . λ:(:S. P→α).y : ∃:S. P ∃-intro-sec

(2)  (λα: ∗ . λ: . . . .y) : ⊥ (pp) on (c), (1)

(3) λ:Py.  (. . . ) : ¬(Py) (bst) on (2)

(4) λy:S. λ:Py.  (. . . ) : ∀y:S.¬(Py) (bst) on (3)

(5) λ:¬(∃ . . . ). λy:S. λ:Py.  (. . . ) : ¬(∃:S. P)⇒ ∀y:S.¬Py (bst) on (4)

∀ and ∃ are written for  and the coding from here on. Line (1) is a hidden
∃-introduction: nothing in the term says so. N&G §7.6

The same proof, as one term

-- the proof object of ¬∃x:S. P x ⇒ ∀y:S. ¬P y, line (5) of the derivation
theorem not_ex (S : Type) (P : S → Prop) : ¬(∃ x, P x) → ∀ y, ¬P y :=

fun u y v => u 〈y, v〉

#print not_ex
-- theorem not_ex : ∀ (S : Type) (P : S → Prop), (¬∃ x, P x) → ∀ (y : S), ¬P y :=
-- fun S P u y v => u (Exists.intro y v)

-- with the coding instead of Lean’s Exists, line (1) reappears:
theorem not_ex’ (S : Type) (P : S → Prop) : ¬(Ex2 S P) → ∀ y, ¬P y :=

fun u y v => u (fun _ w => w y v)

line of the derivation in Lean

(1) λα: ∗ . λ: . . . .y 〈y, v〉 the anonymous constructor is ∃-intro-sec
(2)  (. . . ) u 〈y, v〉 (pp)
(3)–(5) fun u y v => (bst) three times

Terms grow, and the rules hide

Two things went wrong in the last derivation, and both get worse with size.

Two remedies. Names: write ∃-n(S, P, y, ) for line (1); next lecture. A
machine for the bookkeeping: goal, flags, holes; the rest of this one.

N&G §7.6, §7.7

Derivations are built from the goal

Written top to bottom, found bottom to top: a goal A→B or :A.B raises a
flag; a hypothesis applied backward turns the goal into its premisses.

Term Finding, by a person with a strategy; no algorithm does it.
N&G §2.6, §7.3, §7.4

Tactic mode is that bookkeeping, mechanised

After by, Lean holds the ? line and shows it as a goal state: flags above the
line, the hole below it.

intro x (bst), read bottom to top the goal :A.B becomes B, with a flag  : A
apply h (pp), read bottom to top with h : A→B, the goal B becomes A
exact t the hole is filled outright t must have the type of the goal N&G Rem. 2.4.7

intro raises a flag, exact fills the hole

theorem diag’ (S : Type) (Q : S → S → Prop) :
(∀ x y : S, Q x y) → ∀ u : S, Q u u := by

intro z -- z : ∀ x y, Q x y ⊢ ∀ u, Q u u
intro u -- z : ∀ x y, Q x y, u : S ⊢ Q u u
exact z u u -- no goals

#print diag’
-- theorem diag’ : ∀ (S : Type) (Q : S → S → Prop),
-- (∀ (x y : S), Q x y) → ∀ (u : S), Q u u :=
-- fun S Q z u => z u u

The comments are the goal state after each line. What by produced is the term L3 wrote
by hand.

apply leaves a hole

A hole has a name, ?m, and apply h fills the hole with h ?m. The new hole is
the next goal.
theorem chain (p q r : Prop) (hpq : p → q) (hqr : q → r) :

p → r := by
intro hp -- ⊢ r term so far: fun hp => ?m
apply hqr -- ⊢ q fun hp => hqr ?m
apply hpq -- ⊢ p fun hp => hqr (hpq ?m)
exact hp -- no goals fun hp => hqr (hpq hp)

#print chain
-- theorem chain : ∀ (p q r : Prop), (p → q) → (q → r) → p → r :=
-- fun p q r hpq hqr hp => hqr (hpq hp)

Each apply is one (pp) read upward: the conclusion is known, a premiss is left to find.

The term grows as the goals close

One tactic per rule

Each connective has a rule to prove it and a rule to use it, and each rule is one
tactic. The table is this lecture, read as a script.

to prove it (introduction) to use h (elimination)

A⇒ B intro a apply h, exact h a
∀:S. P intro x apply h, exact h a
A∧ B constructor, exact 〈a, b〉 obtain 〈a, b〉 := h, h.1, h.2
A∨ B left, right rcases h with a | b
¬A intro a exact h a, exact absurd a h
⊥ — exact h.elim, contradiction
∃:S. P exact 〈w, p〉, exists w obtain 〈w, hw〉 := h
A⇔B constructor h.mp, h.mpr

obtain and rcases raise the flags of an elimination rule: for ∃, the  : S and
 : P of the book’s strategy; for ∨, one goal per case. TPIL §5.1–5.3

∧, ∨, ∃: the rules as tactics

example (A B : Prop) (h : A ∧ B) : B ∧ A := by
obtain 〈a, b〉 := h -- ∧-el: two flags a : A, b : B
constructor -- ∧-in: two goals, B and A
· exact b
· exact a

example (A B : Prop) (h : A ∨ B) : B ∨ A := by
rcases h with a | b -- ∨-el: one goal per case
· right; exact a -- ∨-in2
· left; exact b -- ∨-in1

example (S : Type) (P Q : S → Prop) (h : ∃ x, P x ∧ Q x) : ∃ x, Q x := by
obtain 〈w, _, hq〉 := h -- ∃-el then ∧-el: flags w : S, hq : Q w
exact 〈w, hq〉 -- ∃-in

A bullet · focuses one goal: the flag indentation of a derivation, written with a dot.

Computational Logic in the Era of AI 36 / 43



¬, ⊥, ∀,⇔: the rules as tactics

example (A : Prop) (h : ¬A) (a : A) : False := by exact h a -- ¬-el
example (A : Prop) (a : A) : ¬¬A := by -- ¬-in

intro h; exact h a
example (A : Prop) (h : False) : A := by exact h.elim -- ⊥-el
example (A : Prop) (h : False) : A := by contradiction -- the same

example (S : Type) (P : S → Prop) (h : ∀ x, P x) (a : S) : P a := by
apply h -- ∀-el

example (S : Type) (P : S → Prop) (h : ∀ x, P x ∧ P x) : ∀ y, P y := by
intro y; exact (h y).1 -- ∀-in

example (A B : Prop) (f : A → B) (g : B → A) : A ↔ B := by
constructor -- ⇔-in
· exact f
· exact g

The propositional proof, as a script

theorem or_neg’ (A B : Prop) :
A ∨ B → ¬A → B := by

intro x y -- flags (c), (d)
apply x.elim -- ∨-el, backwards
· intro u -- flag (e)
exact absurd u y -- (2), (3)
· intro v -- flag (f)
exact v -- (6)

#print or_neg’
-- theorem or_neg’ : ∀ (A B : Prop),
-- A ∨ B → ¬A → B :=
-- fun A B x y =>
-- Or.elim x (fun u => absurd u y)
-- fun v => v

Each tactic is one line of the derivation, read
upward: intro raises the flags, apply
x.elim is ∨-elimination with its two
premisses left as goals, and each exact fills a
hole with a term of the derivation.

The term #print shows is the one of the slide
“The same proof, as one term”, with absurd
for the two applications (2), (3).

cases x with | inl u => ...| inr v =>
... spells the same ∨-elimination as case
analysis; its term is the recursor of Or, L7’s
business.

Three views of one proof

The script is the derivation read from the bottom; the goal after each line is the flag
derivation with a ? line; the term is what the kernel checks. One object, three views.

AI checkpoint: search is untrusted, checking is not

A tactic, a library search, a decision procedure and a language model do one
job: propose a term. The kernel checks it, whoever proposed it.

What the check cannot catch
A wrong statement. A checked proof of the wrong proposition is still a checked
proof.

N&G Thm. 6.3.15, §16.2

What is still missing

missing consequence added by

names the coding of ∃ is copied into every
proof that uses it; line (1) of §7.6 hides
an ∃-introduction

definitions (L5)

equality, order, sets everything so far is pure logic;  = y
and  ≤ y are not in the system

equality (L6)

numbers, with
induction

N, 0, s can be assumed, but no coding
gives the induction principle

inductive types (L7)

Next lecture
Definitions: a name is a step of the type theory, the environment is the book,
and a proof step is a definition too. N&G §7.7, §8.1

Chapter 7 of the book throughout: §7.1 absurdity and negation (Rem. 7.1.1, 7.1.2), §7.2
conjunction and disjunction (the derivations of pp. 141 and 143, Rem. 7.2.1), §7.3 the
propositional example (Fig. 7.1, Rem. 7.3.1), §7.4 classical logic, §7.5 the existential
quantifier (Rem. 7.5.1–7.5.3), §7.6 the predicate-logic example, §7.7 the conclusions.
Exercises 7.4, 7.7 and 7.11 are answered on the slides. The Lean slides follow Theorem
Proving in Lean 4: ch. 3 (propositions and proofs), §4.4 (the existential quantifier),
§5.1–5.3 (entering tactic mode, the basic tactics), and §7.3 for And, Or, Exists as
inductive types.

